

초음파 레이더 만들기

프로젝트 소개


아두이노 코드 작성하기

프로세싱 코드 작성하기


1 프로젝트 소개

| 초음파 레이더 만들기

| 프로젝트 소개



1m 이내에 있는 사물을 인식하는 초음파 레이더를 만들어본다.



초음파 레이더 3D 모델
<http://goo.gl/eaqvjq>



A에 서보 모터를 꼽고 나사로 고정시켜준다.



B에 초음파센서를 끼워준다.



B 밑 부분에 글루건을 이용해 서보 모터 축을 붙여준다.



A와 B를 연결한다.

▪ 재료



2 아두이노 코드 작성하기

| 아두이노 코드 작성하기

```
#include <Servo.h>
#define TRIG 2
#define ECHO 3

Servo servo;
int servoDirection = 1, rad = 0;

void setup(){
  Serial.begin(9600);
  pinMode(TRIG, OUTPUT);
  pinMode(ECHO, INPUT);
  servo.attach(9);
}

void loop(){
  digitalWrite(TRIG, LOW);
  delayMicroseconds(2);
  digitalWrite(TRIG, HIGH);
  delayMicroseconds(5);
  digitalWrite(TRIG, LOW);

  long distance = pulseIn(ECHO, HIGH, 5800)/58;

  Serial.print("r");
  Serial.print(rad);
  Serial.print("d");
  Serial.println(distance);

  rad += servoDirection;
  if(rad > 180){
    rad = 179;
    servoDirection = -1;
  }else if(rad < 0){
    rad = 1;
    servoDirection = 1;
  }
  servo.write(rad);
  delay(15);
}
```

2 아두이노 코드 작성하기

▪ pulseIn

```
pulseIn(ECHO, HIGH, 5800)
```

특정 핀이 정해진 상태가 될때까지 걸린 시간을 알려주는 함수

```
pulseIn(ECHO, HIGH, 5800)
```

↑
Microseconds
최대 기다리는 시간으로
기본은 1초(1,000,000)이다.

| 프로세싱 코드 작성하기

```
import processing.serial.*;

Serial myPort;
int rad = 0;
float cx, cy;
ArrayList<Ball> balls = new ArrayList<Ball>();

void setup()
{
    size(400, 400);
    cx = width/2;
    cy = height*2/3;
    noFill();

    println(Serial.list());
    myPort = new Serial(this, Serial.list()[0], 9600);
}

void draw(){
    background(0);
    stroke(0, 255, 36);
    arc(cx, cy, width, height, PI, TWO_PI, CHORD);
    float r = TWO_PI-map(rad, 0, 360, 0, TWO_PI);
    line(cx, cy, cx + cos(r)* width/2, cy + sin(r)* height/2);
    updateBalls();
    displayBalls();
}

void updateBalls(){
    for(int i = balls.size()-1 ; i > -1 ; i--){
        balls.get(i).update();
        if(balls.get(i).isDead())
            balls.remove(i);
    }
}

void displayBalls(){
    for(int i = 0 ; i < balls.size() - 1 ; i++){
        balls.get(i).display();
    }
}
```

```

void serialEvent(Serial p){
  String inString = p.readStringUntil('\n');
  if(inString != null){
    if(inString.startsWith("r")){
      String[] strings = inString.trim().replace("r", "").split("d");
      if(strings.length > 1){
        rad = Integer.parseInt(strings[0]);
        int distance = Integer.parseInt(strings[1]);
        if(distance != 0){
          balls.add(new Ball(cx, cy, rad, distance));
        }
      }
    }
  }
}

```

▪ Ball

```

class Ball{
  int life = 50;
  float x, y;

  public Ball(float cx, float cy, int rad, int distance){
    float d = map(distance, 0, 100, 0, width/2);
    float r = TWO_PI-map(rad, 0, 360, 0, TWO_PI);
    x = cx + cos(r)*d;
    y = cy + sin(r)*d;
  }

  public void display(){
    ellipse(x, y, life, life);
  }

  public void update(){
    --life;
    if(life < 0 )
      life = 0;
  }

  public boolean isDead(){
    return life == 0;
  }
}

```

▪ ArrayList

```
ArrayList<Ball> balls = new ArrayList<Ball>();
```

변수를 자유롭게 넣었다 뺐다 할 수 있는 배열

```
ArrayList<Ball> balls = new ArrayList<Ball>();
```

↑
변수 형태

↑
초기화

▪ ArrayList.size()

```
balls.size()
```

ArrayList의 배열 크기를 반환하는 함수

- ArrayList.get()

```
balls.get(i)
```

ArrayList에서 특정 위치에 있는 자료를 반환하는 함수

```
balls.get(i)
```

↑
위치

- ArrayList.remove()

```
balls.remove(i);
```

ArrayList에서 특정 위치에 있는 자료를 제거하는 함수

```
balls.remove(i);
```

↑
위치

3 프로세싱 코드 작성하기

ArrayList.add()


```
balls.add(new Ball(cx, cy, rad, distance));
```

ArrayList에 자료를 추가하는 함수

```
balls.add(new Ball(cx, cy, rad, distance));
```


↑
추가할 자료

| 확인하기



3 프로세싱 코드 작성하기

| 확인하기



초음파센서가 좌우로 움직이는 것에 맞춰 프로세싱에 레이더 화면도 변한다.